*Full sample - main results for Republican subsample & alternative model specifications

preserve
keep if rep==1

local demog1 female c.age c.age2 i.educat i.inc4 hisp c.perc_hispanic c.perc_hispanic2
local demog2 female c.age c.age2 i.educat i.inc4 hisp i.pct_hisp_all i.pct_asian_all
local econ i.econ_country i.econ_house
local feeling c.ft_hisp c.ft_as
local pol i.pid3 i.ideo3
local X reform_support5 w2 w3
local inter whiteevan w2_whiteevan w3_whiteevan

foreach j of numlist 1 2{
eststo clear
qui reg `X' if whiteevan==1 [aw=finalweight1], cluster(wustlid)
est store a1
qui reg `X'  `inter' if insample_support==1 [aw=finalweight1], cluster(wustlid)
est store a2
qui reg `X'  `inter'  `demog`j'' w2#(`demog`j'') w3#(`demog`j'') if insample_support==1 [aw=finalweight1], cluster(wustlid)
est store a3
qui reg `X'  `inter'  `demog`j'' `econ' w2#(`demog`j'' `econ') w3#(`demog`j'' `econ') if insample_support==1 [aw=finalweight1], cluster(wustlid)
est store a4
qui reg `X'  `inter'  `demog`j'' `econ' `feeling' w2#(`demog`j'' `econ' `feeling') w3#(`demog`j'' `econ' `feeling') if insample_support==1 [aw=finalweight1], cluster(wustlid)
est store a5
qui reg `X'  `inter'  `demog`j'' `econ' `feeling' `pol' w2#(`demog`j'' `econ' `feeling' `pol') w3#(`demog`j'' `econ' `feeling' `pol') if insample_support==1 [aw=finalweight1], cluster(wustlid)
est store a6

esttab a1 a2 a3 a4 a5 a6  using "~/Dropbox/Immigration/JOP Results/TAPS/tables/support_reform_rep_geocode`j'.tex", ///
keep(w2 w3 whiteevan w2_whiteevan w3_whiteevan _cons) ///
star(* .10 ** 0.05) obslast nogaps ///
sfmt(4) se(2) b(2) replace booktabs compress label r2 ///
title("White evangelicals became more supportive of immigration reform over time (alternative specification `j')")  ///
nomtitles ///
coeflabels(w2  "Wave 2" w3 "Wave 3" whiteevan "White evangelical" w2_whiteevan "Wave 2 * White evangelical" w3_whiteevan "Wave 3 * White evangelical" _cons "Intercept") ///
indicate( "Demographics = age" "Perceptions of economy = *econ_country*" "Feelings toward immigrant groups = *ft_hisp*" "Politics = *pid3*") ///
nonotes
}

restore
